```
$TITLE: M9-2.GMS:  Two Country Large-Group Monopolistic Competition

$ontext
 same data is used as used in M9-1, calibrated to monopolistic comp

        YI    YJ    XMI   XMJ   NMI   NMJ    WI    WJ   CONI  CONJ  EHTI ENTJ

PYI    100                                 -100
PYJ          100                                 -100
PXI                 100                   -50   -50
PXJ                       100             -50   -50
FCI                             20                                  -20
FCJ                                   20                                -20
PSI    -40          -48          -12                 100
PSJ           -40          -48          -12                 100
PUI    -60          -32           -8                 100
PUJ           -60          -32           -8                 100
PWI                                       200         -200
PWJ                                             200         -200
MKI                 -10   -10                                     10    10
MKJ                 -10   -10                                     10    10
$offtext

PARAMETERS
 SI          sigma: elasticity of substitution among varieties
 TC          trade costs on a gross basis (TC = 1 is costless trade)
```

```
 FC          fixed costs
 E0          scaling parameter for calibration
 ENDOWIS     endowment of skilled labor in country i
 ENDOWIL     endowment of unskilled labor in country i
 ENDOWJS     endowment of skilled labor in country j
 ENDOWJL     endowment of unskilled labor in country j
 MODELSTAT   indicator whether or not model solved
 REALPUI     real price of unskilled labor in i
 REALPUJ     real price of unskilled labor in j
 REALPSI     real price of skilled labor in i
 REALPSJ     real price of skilled labor in j;

SI = 5;
TC = 1.;
FC = 20;
ENDOWIS = 1;
ENDOWIL = 1;
ENDOWJS = 1;
ENDOWJL = 1;

* E0: scaling parameter s.t. the consumer price index PW = 1 initially

E0 = (1.25**(1-SI) + 1.25**(1-SI))**(1/(1-SI));
DISPLAY E0;
```

POSITIVE VARIABLES

```
 WFI    welfare of country i
 WFJ    welfare of country j
 XII    production of X in i for sale in i
 XIJ    production of X in i for sale in j
 XJJ    production of X in j for sale in j
 XJI    production of X in j for sale in i
 YI     production of Y in country i
 YJ     production of Y in country j
 NI     number of national firms in i (number of "varieties")
 NJ     number of national firms in j
 PXI    price of an X variety in country i
 PXJ    price of an X variety in country j
 PY     price of Y: domestic and world (no trade costs)
 PWI    price of welfare (real consumer price index) in i
 PWJ    price of welfare (real consumer price index) in j
 PEI    price index for the X composite good in i
 PEJ    price index for the X composite good in j
 PSI    price of skilled labor in i
 PUI    price of unskilled labor in i
 PSJ    price of skilled labor in j
 PUJ    price of unskilled labor in j
 CONSI  consumer income in i
 CONSJ  consumer income in j;
```

```
EQUATIONS
 PRWI    pricing equation for WI
 PRWJ    pricing equation for WJ
 PRXI    MC gte MR for X produced in i (same for all firm types)
 PRXJ    MC gte MR for X produced in j (same for all firm types)
 PRYI    MC gte PY for Y produced in i
 PRYJ    MC gte PY for Y produced in j
 PRFI    MC gte PFI for fixed costs in i
 PRFJ    MC gte PFJ for fixed costs in j
 DXII    supply-demand for a X variety produced in i and sold in i
 DXJI    supply-demand for a X variety produced in j and sold in i
 DXJJ    supply-demand for a X variety produced in j and sold in j
 DXIJ    supply-demand for a X variety produced in i and sold in j
 DY      supply-demand for world production and consumption of Y
 DWI     supply-demand for welfare in i
 DWJ     supply-demand for welfare in j
 PINDEXI price index for the X composite in i
 PINDEXJ price index for the X composite in j
 SKLABI  supply-demand for skilled labor in i
 UNLABI  supply-demand for unskilled labor in i
 SKLABJ  supply-demand for skilled labor in j
 UNLABJ  supply-demand for unskilled labor in j
 ICONSI  income-expenditure balance in i
 ICONSJ  income-expenditure balance in j;
```

```
PRWI..    ((PEI/E0)**0.5)*(PY**0.5) =G= PWI;

PRWJ..    ((PEJ/E0)**0.5)*(PY**0.5) =G= PWJ;

PRXI..    (PUI**0.4)*(PSI**0.6) =G= PXI*(1-1/SI);

PRXJ..    (PUJ**0.4)*(PSJ**0.6) =G= PXJ*(1-1/SI);

PRYI..    (PUI**0.60)*(PSI**0.40)  =G= PY;

PRYJ..    (PUJ**0.60)*(PSJ**0.40)  =G= PY;

PRFI..    FC*(SI-1) =G= XII*40 + XIJ*40;

PRFJ..    FC*(SI-1) =G= XJJ*40 + XJI*40;

DXII..    XII*40 =E= PXI**(-SI)*(PEI**(SI-1))*CONSI/2;

DXJI..    XJI*40/TC =E= (PXJ*TC)**(-SI)*(PEI**(SI-1))*CONSI/2;

DXJJ..    XJJ*40 =E= PXJ**(-SI)*(PEJ**(SI-1))*CONSJ/2;

DXIJ..    XIJ*40/TC =E= (PXI*TC)**(-SI)*(PEJ**(SI-1))*CONSJ/2;

DY..      YI*100 + YJ*100 =E= CONSI/(2*PY) + CONSJ/(2*PY);
```

```
DWI..     200*WFI =E= CONSI/(PWI);

DWJ..     200*WFJ =E= CONSJ/(PWJ);

PINDEXI.. PEI =E= (NI*PXI**(1-SI) + NJ*(PXJ*TC)**(1-SI))**(1/(1-SI));

PINDEXJ.. PEJ =E= (NI*(PXI*TC)**(1-SI) + NJ*PXJ**(1-SI))**(1/(1-SI));


SKLABI..  100*ENDOWIS =E= 0.40*YI*100*PY/PSI
              + 0.6*NI*((XII+XIJ)*40 + FC)*PXI*(1-1/SI)/PSI;

UNLABI..  100*ENDOWIL =E= 0.60*YI*100*PY/PUI
              + 0.4*NI*((XII+XIJ)*40 + FC)*PXI*(1-1/SI)/PUI;

SKLABJ..  100*ENDOWJS =E= 0.40*YJ*100*PY/PSJ
              + 0.6*NJ*((XJJ+XJI)*40 + FC)*PXJ*(1-1/SI)/PSJ;

UNLABJ..  100*ENDOWJL =E= 0.60*YJ*100*PY/PUJ
              + 0.4*NJ*((XJJ+XJI)*40 + FC)*PXJ*(1-1/SI)/PUJ;

ICONSI..  CONSI =E= PSI*100*ENDOWIS + PUI*100*ENDOWIL;

ICONSJ..  CONSJ =E= PSJ*100*ENDOWJS + PUJ*100*ENDOWJL;
```

```
MODEL M9_2 /PRWI.WFI, PRWJ.WFJ, PRXI.PXI, PRXJ.PXJ, PRYI.YI, PRYJ.YJ,
       PRFI.NI, PRFJ.NJ,
       DXII.XII, DXJI.XJI, DXJJ.XJJ, DXIJ.XIJ, DY.PY, DWI.PWI, DWJ.PWJ,
       PINDEXI.PEI, PINDEXJ.PEJ,
       SKLABI.PSI, SKLABJ.PSJ, UNLABI.PUI, UNLABJ.PUJ,
       ICONSI.CONSI, ICONSJ.CONSJ/;

OPTION MCP=PATH;

WFI.L = 1;
WFJ.L = 1;
PWI.L = 1;
PWJ.L = 1;
PEI.L = E0;
PEJ.L = E0;
CONSI.L = 200;
CONSJ.L = 200;
XII.L = 1;
XIJ.L = 1;
XJJ.L = 1;
XJI.L = 1;
YI.L = 1;
YJ.L = 1;
NI.L = 1;
NJ.L = 1;
PXI.L = 1.25;
```

```
PXJ.L = 1.25;
PY.L = 1;
PSI.L = 1;
PUI.L = 1;
PSJ.L = 1;
PUJ.L = 1;

PY.FX = 1;
TC = 1.;

SOLVE M9_2 USING MCP;

MODELSTAT = M9_2.MODELSTAT - 1.;

* counterfactual: trade costs of 10%

TC = 1.1;

SOLVE M9_2 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ;
```

```
* counterfactual: country's identical except for size,
* positive trade costs (home market advantage)

TC = 1.1;
ENDOWIL = 1.5;
ENDOWIS = 1.5;
ENDOWJL = 0.5;
ENDOWJS = 0.5;

SOLVE M9_2 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ;

* counterfactual: country h has a comparative advantage in X
* example of unstable symmetric equilibrium
* under factor mobility: both factor's real prices higher in I

TC = 1.1;
ENDOWIL = 0.80;
ENDOWIS = 1.20;
```

```
ENDOWJL = 1.20;
ENDOWJS = 0.80;

SOLVE M9_2 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ;

* counterfactual: country h has a comparative advantage in X
* no trade costs

TC = 1.;
ENDOWIL = 0.80;
ENDOWIS = 1.20;
ENDOWJL = 1.20;
ENDOWJS = 0.80;

SOLVE M9_2 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
```

```
REALPSJ = PSJ.L/PWJ.L;

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ;
```